Widerstand berechnen

$$U_R = U \cdot \frac{R}{10 + R}$$

$$\frac{U_R}{U} = \frac{R}{R\left(\frac{10}{R} + 1\right)}$$

$$\frac{U_R}{U} = \frac{1}{\frac{10}{R} + 1}$$

$$\frac{10}{R} = \frac{1}{\frac{U_R}{U}} - 1$$

$$\frac{10}{\frac{1}{\frac{U_R}{U}} - 1} = R$$

$$\frac{10 \cdot U_R}{U - U_B} = R$$

Formel für NTC Temperatur

$$R_\mathrm{T} = R_\mathrm{N} \cdot \mathrm{e}^{B\left(\frac{1}{T}-\frac{1}{T_\mathrm{N}}\right)} \quad \Leftrightarrow \quad \frac{1}{T} = \frac{1}{T_\mathrm{N}} + \frac{1}{B}\ \ln\frac{R_\mathrm{T}}{R_\mathrm{N}}$$

von https://de.wikipedia.org/wiki/Hei%C3%9Fleiter

$$T = \frac{1}{\frac{1}{T_N} + \frac{1}{B}\ln\frac{R}{R_N}}$$

$$T_N = 298{,}15$$

$$B = 4000$$

$$R_N = 10000$$

$$T = \frac{1}{\frac{1}{T_N} + \frac{\ln\frac{R}{R_N}}{B}}$$

$$T = \frac{1}{\frac{T_N \cdot \ln\frac{R}{R_N} + B}{T_N \cdot B}}$$

$$T = \frac{T_N \cdot B}{T_N \cdot \ln\frac{R}{R_N} + B}$$

```ino
// Programm zum auslesen von einem NTC, LDR und Poti
// Schaltplan und mehr Infos in Moodle unter GYT26_TIF_12_SM

// Pin-Belegung (A: analog)
#define A_NTC A0
#define A_LDR A1
#define A_POTI A2

#define KELVIN_CONSTANT 273.15f

uint16_t iteration = 0;

void setup()
{
  Serial.begin(9600);
}

// Sensor Spannung in mV analog lesen
float read_voltage(const uint8_t analog_pin)
{
  // (Analog Wert * Gesamt Spannung) / Maximaler Analog Wert (10 bit)

  return analogRead(analog_pin) * 5000.0f / 1023.0f;
}

// Sensor Widerstand in Ohm lesen
float read_resistance(const uint16_t sensor_mv)
{
  /*
  Vorwiderstand * Sensor Spannung / (Gesamtspannung - Sensor Spannung)

  Hergeleitet aus:
    Sensor Spannung = Gesamt Spannung * Sensor Widerstand
                    / (Vorwiderstand + Sensor Widerstand)
  */

  return 10000.0f * sensor_mv / (5000.0f - sensor_mv);
}
```

```c
/* NTC Temperatur in Gradcelsius lesen
  Standart werte:
  ambient_temperature_kelvin = 298.15f (25 degree celsius)
  ambient_resistance_ohm = 10000.0f (10k Ohm)
  b_constant = 4000.0f
*/
float read_temperature(const float sensor_ohm, const float ambient_temperature_kelvin,
  const float ambient_resistance_ohm, const float b_constant)
{
  // Formel aus https://de.wikipedia.org/wiki/Hei%C3%9Fleiter
  const float temperature_kelvin = (ambient_temperature_kelvin * b_constant)
    / (ambient_temperature_kelvin * log(sensor_ohm / ambient_resistance_ohm) + b_constant);

  return temperature_kelvin - KELVIN_CONSTANT;
}

// LDR Helligkeit in Lux lesen
uint16_t read_brightness(const uint16_t sensor_ohm)
{
  // Annäherung:
  return 500 * sensor_ohm;
}

// Potiwiderstand in mm lesen
uint16_t read_distance(const uint16_t sensor_ohm)
{
  // 10cm = 100mm
  // 1000 Ohm <=> 100mm | : 100
  // => 10 Ohm <=> 1mm
  return sensor_ohm / 10;
}
```

```
void loop()
{
  const uint16_t ntc_mv = read_voltage(A_NTC);
  const uint16_t ldr_mv = read_voltage(A_LDR);
  const uint16_t poti_mv = read_voltage(A_POTI);

  Serial.print("Messung: ");
  Serial.print(iteration);
  Serial.print("\tNTC: ");
  Serial.print(ntc_mv);
  Serial.print(" mv\tLDR: ");
  Serial.print(ldr_mv);
  Serial.print(" mv\tPoti: ");
  Serial.print(poti_mv);
  Serial.println(" mv");

  const uint16_t ntc_ohm = read_resistance(ntc_mv);
  const uint16_t ldr_ohm = read_resistance(ldr_mv);
  const uint16_t poti_ohm = read_resistance(poti_mv);

  Serial.print("Messung: ");
  Serial.print(iteration);
  Serial.print("\tNTC: ");
  Serial.print(ntc_ohm);
  Serial.print(" Ohm\tLDR: ");
  Serial.print(ldr_ohm);
  Serial.print(" Ohm\tPoti: ");
  Serial.print(poti_ohm);
  Serial.println(" Ohm");

  const uint16_t ntc_temperature = read_temperature(ntc_ohm, 298.15f, 10000.0f, 4000.0f);
  const uint16_t ldr_brightness = read_brightness(ldr_ohm);
  const uint16_t poti_distance = read_distance(poti_ohm);

  Serial.print("Messung: ");
  Serial.print(iteration);
  Serial.print("\tNTC: ");
  Serial.print(ntc_temperature);
  Serial.print(" °C\tLDR: ");
  Serial.print(ldr_brightness);
  Serial.print(" Lux\tPoti: ");
  Serial.print(poti_distance);
  Serial.println(" mm");

  Serial.println("");

  delay(500);
  iteration++;
}
```

Output Serial Monitor ×

Message (Enter to send message to 'Arduino Uno' on '/dev/ttyACM0')

```
Messung: 229    NTC: 1246 mv    LDR: 1290 mv    Poti: 1192 mv
Messung: 229    NTC: 3319 Ohm   LDR: 3477 Ohm   Poti: 3130 Ohm
Messung: 229    NTC: 51 °C      LDR: 34564 Lux  Poti: 313 mm

Messung: 230    NTC: 1246 mv    LDR: 1295 mv    Poti: 1192 mv
Messung: 230    NTC: 3319 Ohm   LDR: 3495 Ohm   Poti: 3130 Ohm
Messung: 230    NTC: 51 °C      LDR: 43564 Lux  Poti: 313 mm

Messung: 231    NTC: 1241 mv    LDR: 1290 mv    Poti: 1192 mv
Messung: 231    NTC: 3301 Ohm   LDR: 3477 Ohm   Poti: 3130 Ohm
Messung: 231    NTC: 51 °C      LDR: 34564 Lux  Poti: 313 mm

Messung: 232    NTC: 1246 mv    LDR: 1290 mv    Poti: 1192 mv
Messung: 232    NTC: 3319 Ohm   LDR: 3477 Ohm   Poti: 3130 Ohm
Messung: 232    NTC: 51 °C      LDR: 34564 Lux  Poti: 313 mm
```